﻿
CREATE PROCEDURE [dbo].[LedgerAccountsUpdate]
(
	@ParentId int,	--ссылка на родительский счет (для субсчета)
	@AccountNumber varchar(50), --счет
	@Name varchar(100),
	@Note varchar(250),
	@MD datetime,
	@MW varchar(50),
	@ToDelete bit,
	@Original_Id int,
	@Original_MD datetime
)
AS
DECLARE @strMsg nvarchar(500), @intRet int, @intErrCode int
DECLARE @orgToDelete bit
SET NOCOUNT ON

IF (@Name Is Null) OR (@AccountNumber Is Null) OR (@ToDelete Is Null)
OR (@Original_Id Is Null) OR (@Original_MD Is Null)
BEGIN
	SELECT @intErrCode = 1, @strMsg = 'Неправильные параметры в вызове процедуры ' + Object_Name(@@Procid) + ' !'
	RAISERROR (@strMsg, 11, @intErrCode)
	RETURN @intErrCode
END

IF (ltrim(@Name) = '')  OR (ltrim(@AccountNumber) = '') 
BEGIN
	SELECT @intErrCode = 2, @strMsg = 'Незаполнено наименование или номер счета!'
	RAISERROR (@strMsg, 11, @intErrCode)
	RETURN @intErrCode
END

IF @ParentId Is Not Null
BEGIN
	IF NOT EXISTS(SELECT Id FROM CO.LedgerAccounts WHERE Id = @ParentId AND ToDelete = 0)
	BEGIN
		SELECT @intErrCode = 3, 
		@strMsg = 'Отсутствует бухгалтерский счет с идентификатором ' + cast(@ParentId as varchar(20)) + '!'
		RAISERROR (@strMsg, 11, @intErrCode)
		RETURN @intErrCode
	END
END

SELECT @orgToDelete = ToDelete
FROM [CO].LedgerAccounts WHERE Id = @Original_Id AND MD = @Original_MD
IF (@@rowcount <> 1)
BEGIN
	SELECT @intErrCode = 3, 
	@strMsg = 'Данные изменены после последнего запроса!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF @orgToDelete = 1 --Запись была отмечена для удаления
BEGIN
	IF @ToDelete = 1 
	BEGIN
		SELECT @intErrCode = 4, 
		@strMsg = 'Нельзя редактировать запись, отмеченную для удаления!'
		RAISERROR (@strMsg, 11, 1)
		RETURN @intErrCode
	END
END ELSE BEGIN
	IF @ToDelete = 1 
	BEGIN
		SELECT @intErrCode = 5, 
		@strMsg = 'Для удаления данных должна использоваться другая процедура!'
		RAISERROR (@strMsg, 11, 1)
		RETURN @intErrCode
	END
END

EXEC [dbo].[ProcessStatistic] @Note OUTPUT, @MD OUTPUT, @MW OUTPUT

	SET NOCOUNT OFF;

UPDATE [CO].LedgerAccounts SET [ParentId] = @ParentId, 
AccountNumber = @AccountNumber,
[Name] = @Name,
[Note] = @Note, [MD] = @MD, [MW] = @MW, ToDelete = @ToDelete 
WHERE Id = @Original_Id AND MD = @Original_MD;

SELECT Id, ParentId, AccountNumber, Name, Note, MD, MW, ToDelete FROM CO.LedgerAccounts
WHERE (Id = @Original_Id)

RETURN 0
